

ŠEST ZADATAKA RIJEŠENIH PROGRAMOM MAPLE V


ZVONKO ČERIN, ZAGREB

SAŽETAK. U ovom prvom nastavku članka *Analitička geometrija ravnine računalom* od M. Bator, Z. Čerina i M. Čulav iz prošlog broja Matematičko-fizičkog lista detaljno se opisuju rješenja šest zadataka iz zbirke za prvi razred gimnazije u računalnom programu Maple V.

U članku [1] opisano je kako se na računalu u programu Maple V uvodi analitička geometrija ravnine. Kao ilustraciju primjene tog načina rješavanja zadataka dali smo samo tri primjera iz zbirke [2]. U ovom prvom nastavku opisat ćemo još šest primjera koje ponovo biramo iz spomenute zbirke za prvi razred gimnazije. Nadamo se da će vas ovi primjeri potaknuti da i sami pokušate riješiti slične zadatke na računalu.

Započinjemo sa zadacima 719. i 720. iz knjige [2].

Zadatak 1. Dokaži: ako su u trokutu dvije visine ili dvije težišnice jednake, onda je trokut jednakokračan.



SLIKA 1. Ako je $|AA'| = |BB'|$ ili $|AA''| = |BB''|$ onda je $|BC| = |CA|$.

Rješenje. Uz pretpostavke i oznake iz rješenja zadatka 3 u [1], nared-

$\text{FS}(\text{udaljenost2t}(\mathbf{eA}, \mathbf{eApp})^2 - \text{udaljenost2t}(\mathbf{eB}, \mathbf{eBpp})^2);$

dobivamo $\frac{c^3 v^2 (2u-c)}{(v^2+u^2-2uc+c^2)(v^2+u^2)}$. Dakle, ako su visine AA'' i BB'' jednake duljine mora biti $u = \frac{c}{2}$ pa je ABC jednakokračan trokut jer vrh C leži na simetrali stranice AB .


Slično se vidi da unosom u program Maple V

$\text{FS}(\text{udaljenost2t}(\mathbf{eA}, \mathbf{eAp})^2 - \text{udaljenost2t}(\mathbf{eB}, \mathbf{eBp})^2);$

dobijemo $\frac{3c(2u-c)}{4}$ što vodi na isti zaključak za težišnice. \square

Mnogo složeniji za dokazati je zadatak 721 iz [2]. Naša metoda dokaza pretpostavlja poznavanje trigonometrijskih funkcija (posebno kotangensa).

Zadatak 2. Dokaži da je trokut jednakokračan onda i samo onda ako su mu dvije simetrale kutova jednake.



SLIKA 2. Stranice BC i CA su jednake onda i samo onda ako su simetrale AA_1 i BB_1 kutova A i B jednake.

Rješenje. Da bismo dobili što je moguće jednostavnije izraze pretpostaviti ćemo da točke A i B te središte I upisane kružnice trokuta ABC imaju koordinate $(0, 0)$, $(f+g, 0)$ i $(f, 1)$ gdje su f i g pozitivni realni brojevi. To su kotangensi polovica kutova A i B .

$\mathbf{tA} := [0, 0]; \mathbf{tB} := [f+g, 0]; \mathbf{tI} := [f, 1]; \mathbf{tJc} := [f, 0];$

Ako su J_a, J_b, J_c projekcije središta I upisane kružnice na stranice, onda J_c ima koordinate $(f, 0)$ a koordinate od J_a dobivamo iz sistema jednadžbi

$\mathbf{rJa} := \text{solve}(\{\text{udaljenost2t}(\mathbf{tB}, [\mathbf{p}, \mathbf{q}]) = \text{udaljenost2t}(\mathbf{tB}, \mathbf{tJc}), \text{udaljenost2t}(\mathbf{tI}, [\mathbf{p}, \mathbf{q}]) = 1\}, \{\mathbf{p}, \mathbf{q}\}); \mathbf{tJa} := \text{subs}(\mathbf{rJa}[2], [\mathbf{p}, \mathbf{q}]);$

gdje su p i q tražene koordinate točke J_a . Taj sistem ima dva rješenja i to koordinate točke J_c i koordinate $\frac{f(g^2+1)+2g}{g^2+1}$ i $\frac{2g^2}{g^2+1}$ od J_a . Na sličan način se odrede i koordinate $\frac{f(f^2-1)}{f^2+1}$ i $\frac{2f^2}{f^2+1}$ točke J_b .

```
rJb:=solve({udaljenost2t(tA,[p,q])=udaljenost2t(tA,tJc),
udaljenost2t(tI,[p,q])=1},{p,q}); tJb:=subs(rJb[2],[p,q]);
```

Sada možemo odrediti točke A_1 i B_1 u kojima simetrale kutova A i B sijeku nasuprotne stranice kao presjeke $AI \cap BJ_a$ i $BI \cap AJ_b$.

```
tA1:=presjek2p(pravac2t(tA,tI),pravac2t(tB,tJa));
tB1:=presjek2p(pravac2t(tB,tI),pravac2t(tA,tJb));
```

Zatražimo li od programa Maple V da izračuna razliku kvadrata dužina simetrala kuteva utipkavanjem naredbe

```
Q:=FS(udaljenost2t(tA,tA1)^2-udaljenost2t(tB,tB1)^2);
```

kao izračun dobivamo

$$\frac{4(f+g)^3(f-g)(f^4g^2+4g^3f^3-5f^2g^2+g^4f^2+4fg-1)}{(g^2+2fg-1)^2(f^2+2fg-1)^2}.$$

Budući da brojnik sadrži $f-g$ kao faktor i $f+g$ očito nikada nije nula, dokaz će biti gotov ako pokažemo da je

$$Z = f^4g^2 + 4g^3f^3 - 5f^2g^2 + g^4f^2 + 4fg - 1$$

uvijek pozitivan.

Prvo primijetimo da je zbroj $\frac{A}{2} + \frac{B}{2}$ polovica kuteva najviše $\frac{\pi}{2}$ pa je

$$\cot\left(\frac{A}{2} + \frac{B}{2}\right) = \frac{\cot(\frac{A}{2})\cot(\frac{B}{2}) - 1}{\cot(\frac{A}{2}) + \cot(\frac{B}{2})} = \frac{fg - 1}{f + g} > 0$$

što vodi na zaključak da je $fg > 1$.

Prvi i četvrti član od Z zajedno daju

$$f^4g^2 + f^2g^4 = (f^2 + g^2)(fg)^2 \geq 2(fg)(fg)^2 = 2(fg)^3$$

jer je $f^2 + g^2 \geq 2fg$. Uvedemo li oznaku $\vartheta = fg$ vidimo da je

$$Z \geq 6\vartheta^3 - 5\vartheta^2 + 4\vartheta - 1.$$

Kako je $\vartheta > 1$ zamijenimo li ϑ u gornjem kubnom polinomu s $1 + \vartheta$ dobiti ćemo $(3\vartheta + 2)(2\vartheta^2 + 3\vartheta + 2)$ što je doista uvijek pozitivno jer je nova varijabla ϑ pozitivna pa je dokaz gotov.

```
t:=theta: P:=6*t^3-5*t^2+4*t-1: FS(subs(t=t+1,P));
```

Primijetimo da smo to isto mogli postići tako da u polinomu Z stavimo zamjenu $f = \frac{1+k}{g}$ gdje je k pozitivan broj. Na naredbu

```
collect(subs(f=(1+k)/g, op(4,Q)), g);
```


program Maple V izračuna izraz

$$(1+k)^2g^2 + \frac{(1+k)^4}{g^2} + 2 + 6k + 7k^2 + 4k^3$$

koji je, očigledno, pozitivan. □

Nastavljamo zadatkom 833 iz [2] iz odjeljka o sličnosti trokuta.

Zadatak 3. Neka je r polumjer trokutu ABC upisane, a R polumjer opisane kružnice. Dokaži da je $R \geq 2r$.



SLIKA 3. Radijus opisane kružnice je veći ili jednak promjeru upisane kružnice.

Rješenje. Sljedeća metoda rješavanja ovog zadatka ima veliku sličnost s rješavanjem prethodnog zadatka. Bez smanjenja općenitosti možemo pretpostaviti da su kutevi A i B promatranog trokuta oštri (tj. manji od $\frac{\pi}{2}$ radijana) i da vrhovi A , B i središte I upisane kružnice imaju koordinate $(0, 0)$, $(r(f+g), 0)$ i (fr, r) za realne brojeve $f > 1$, $g > 1$ i $r > 0$.

Ideja dokaza je da nađemo koordinate vrha C i središta O opisane kružnice što će nam omogućiti da izračunamo radijus R opisane kružnice i da onda pokažemo da je razlika $R - 2r$ uvijek veća od nule osim u slučaju jednakostraničnog trokuta kada je jednaka nuli.

Ako su J_a , J_b , J_c projekcije središta I upisane kružnice na stranice onda J_c ima koordinate $(fr, 0)$ a koordinate od J_a dobivamo iz sistema jednadžbi

$$\begin{aligned} rJ_a := & \text{solve}(\{\text{udaljenost}^2(tB, [p, q]) = \text{udaljenost}^2(tB, tJc), \\ & \text{udaljenost}^2(tI, [p, q]) = r^2\}, \{p, q\}); \quad tJ_a := \text{subs}(rJ_a[2], [p, q]); \end{aligned}$$

gdje su p i q tražene koordinate točke J_a . Taj sistem ima dva rješenja i to koordinate točke J_c i koordinate $\frac{(g^2f+2g+f)r}{1+g^2}$ i $\frac{2rg^2}{1+g^2}$ od J_a . Na sličan način se odrede i koordinate $\frac{f(f^2-1)r}{f^2+1}$ i $\frac{2f^2r}{f^2+1}$ točke J_b .

$$\begin{aligned} rJ_b := & \text{solve}(\{\text{udaljenost}^2(tA, [p, q]) = \text{udaljenost}^2(tA, tJc), \\ & \text{udaljenost}^2(tI, [p, q]) = r^2\}, \{p, q\}); \quad tJ_b := \text{subs}(rJ_b[2], [p, q]); \end{aligned}$$

Sada možemo odrediti točku C kao presjek $AJ_b \cap BJ_a$.

```
tC:=presjek2p(pravac2t(tA,tJb),pravac2t(tB,tJa));
```

Središte O opisane kružnice i njen radijus R dobivaju se iz sljedećeg sistema.

```
t0:=[p, q]; solve({udaljenost2t(tA,t0)=R,
udaljenost2t(tB,t0)=R, udaljenost2t(tC,t0)=R},{p, q, R});
```

Od dva rješenja samo ono kod kojeg je $R = \frac{r(1+g^2)(1+f^2)}{4(fg-1)}$ zadovoljava. U drugom rješenju je R negativan što nemožemo prihvatiti.

```
R:=r*(1+f^2)*(1+g^2)/4/(f*g-1);
M:=collect(op(2,FS(R-2*r)),f);
Delta:=FS(coeff(M,f,1)^2-4*coeff(M,f,2)*coeff(M,f,0));
```

Razlika $R - 2r$ jednaka je $\frac{Mr}{4(fg-1)}$, gdje je M kvadratni trinom

$$(g^2 + 1)f^2 - 8gf + g^2 + 9$$

u f . Njegova diskriminanta Δ je $-4(-3 + g^2)^2$ što je uvijek negativno (pa je $M > 0$ jer je $g^2 + 1 > 0$) osim kada je $g = \cot \frac{B}{2} = \sqrt{3}$ i zato $f = \sqrt{3}$ (tj. trokut ABC je jednakostraničan) kada je $M = 0$.

□

Naredni je zadatak 312. iz [2] koji je iz drugog poglavlja u dijelu o duljini kružnice i površini kruga.

Zadatak 4. Unutar trokuta ABC odabrana je točka T i neka su A_1, B_1, C_1 redom bilo koje tri unutarnje točke na stranicama $\overline{BC}, \overline{CA}, \overline{AB}$ tog trokuta. Neka su dalje $R_i, i = 1, 2, 3, 4, 5, 6$ redom polumjeri kružnica opisanih trokutima $AC_1T, C_1BT, BA_1T, A_1CT, CB_1T, B_1AT$. Dokaži da vrijedi $R_1 R_3 R_5 = R_2 R_4 R_6$.

Rješenje. Prvo potražimo funkciju koja će danoj trojci točaka pridruživati radijus opisane kružnice trokuta određenog tim točkama.

```
simetrala:=(a, b)-> okomica(poloviste(a, b),pravac2t(a, b));
S0:=(a, b, c)-> presjek2p(simetrala(a, b),simetrala(a, c));
R0:=(a, b, c)-> udaljenost2t(a, S0(a, b, c));
```


Odredimo sada točke A, B, C , i T .

```
tA:=[0, 0]; tB:=[c, 0]; tC:=[s, t]; tT:=[p, q];
```

Ako je $s \neq c$ onda se položaj bilo koje točke A_1 na pravcu BC može odrediti nekim realnim brojem u i koordinate te točke su $\left(u, \frac{t(c-u)}{c-s}\right)$.

To dobijemo tako da tražimo da točka s koordinatama (u, z) leži na pravcu BC i onda taj uvjet riješimo po varijabli z . Slično za $s \neq 0$, bilo koja točka B_1 na pravcu CA ima koordinate $\left(v, \frac{tv}{s}\right)$ a bilo koja točka C_1 na pravcu AB ima koordinate $(w, 0)$ za neke realne brojeve v i w .

```
tA1:=[u, t*(c-u)/(c-s)]; tB1:=[v, t*v/s]; tC1:=[w, 0];
```



SLIKA 4. Za radijuse šest opisanih kružnica vrijedi $R_1 R_3 R_5 = R_2 R_4 R_6$.

Ako je $s = c$ onda bilo koja točka A_2 pravca BC ima koordinate (c, u) za neki realni broj u . Ako je $s = 0$ onda bilo koja točka B_2 pravca CA ima koordinate $(0, v)$ za neki realni broj v .

$tA2 := [c, u]$; $tB2 := [0, v]$;

Odredimo sada funkciju koja računa razliku kvadrata produkata radijusa opisanih kružnica u odnosu na sedam točaka u ravnini.

$FR := (a, b, c, d, e, f, g) \rightarrow FS((RO(a, f, g) * RO(b, d, g) * RO(c, e, g))^2 - (RO(f, b, g) * RO(d, c, g) * RO(e, a, g))^2)$;

Sada se lagano provjeri da su sljedeće vrijednosti jednake nuli:

$FR(tA, tB, tC, tA1, tB1, tC1, tT)$;
 $s := c$; $FR(tA, tB, tC, tA2, tB1, tC1, tT)$;
 $s := 0$, $FR(tA, tB, tC, tA1, tB2, tC1, tT)$;


Tako smo zadatak 312 iz [2] riješili u programu Maple V. □

Napomena 1. Iz gornjeg dokaza je jasno da nigdje nismo koristili pretpostavku da je točka T unutar trokuta ABC niti pretpostavku da su točke A_1, B_1, C_1 unutarnje točke na stranicama $\overline{BC}, \overline{CA}, \overline{AB}$. Na taj način smo upotrebom računala uspjeli dokazati mnogo općenitiju tvrdnju.

Sljedeći primjer je zadatak 644 iz zbirke [2] koji je iz odjeljka o volumenu valjka, stošca i kugle.

Zadatak 5. Na dnu valjkaste posude promjera 15 cm nalazi se kugla promjera 12 cm. U posudu je nalivena voda tako da njen nivo dotiče

najvišu točku kugle. Za koliko će cm pasti nivo vode u posudi ako se kugla izvadi?



SLIKA 5. Valjkasta čaša s kuglom u njoj.

Rješenje. Prisjetimo se formule $V_K = \frac{4}{3} \left(\frac{D}{2}\right)^3 \pi$, za volumen kugle promjera D , i formule $V_V = \left(\frac{d}{2}\right)^2 h \pi$, za volumen valjka kojem je baza krug promjera d a visina h .

U programu Maple V funkcije volumena se definiraju ovako:

```
Vk:=unapply(d^3*Pi/6, d); Vv:=unapply(d^2*h*Pi/4, [d, h]);
```

Volumen vode u čaši je razlika volumena valjka i kugle.

```
Vvoda:=Vv(15, 12)-Vk(12);
```


Nakom vađenja kugle voda se smjesti u valjak kojem je baza krug promjera 15 cm a visina mu je $12 - p$ cm gdje je p traženi pad nivoa vode u čaši. Zato se p nađe u programu Maple V ovako:

```
solve(Vvoda=Vv(15, 12-p), p);
```

Dobije se vrijednost $p = 5.12$ cm. □

Još jedan lijepi primjer je zadatak 963 iz [2]. I ovdje se pretpostavlja poznavanje trigonometrijskih funkcija.

Zadatak 6. Oko kružnice radijusa R opisan je trapez. Tetiva koja spaja dirališta na krakovima paralelna je osnovicama i ima duljinu b . Dokaži da je površina trapeza $P = \frac{8R^3}{b}$.



SLIKA 6. Površina trapeza $ACEG$ je $\frac{8R^3}{b}$.

Rješenje. Odaberimo pravokutni koordinatni sustav tako da kružnica k radijusa R koja je upisana promatranom trapezu $ACEG$ ima središte u ishodištu a njegove paralelne stranice \overline{AC} i \overline{EG} dodiruju k u točkama $B(0, -R)$ i $F(0, R)$. Neka krakovi \overline{CE} i \overline{AG} dodiruju k u točkama $D(R \cos \theta, R \sin \theta)$ i $H(R \cos \sigma, R \sin \sigma)$ za neke kuteve θ i σ .

Zadajmo u programu Maple V prvo točke O, B, F, D, H i pravce AC, EG .

```
t0:=[0, 0]; tB:=[0, -R]; tF:=[0, R];
tD:=[R*cos(theta), R*sin(theta)]; pAC:=[0, 1, R];
tH:=[R*cos(sigma), R*sin(sigma)]; pEG:=[0, 1, -R];
```

Prvo tražimo kada će spojnica \overline{DH} dirališta krakova biti paralelna s osnovicama.

```
paralelniQ(pravac2t(tD, tH), pAC);
```

Uvjet je $R(\sin \theta - \sin \sigma) = 0$ pa mora biti $\sigma = \pi - \theta$. Dakle, trapez $ACEG$ je istokračan i simetričan u odnosu na pravac BF pa je dovoljno odrediti površinu desne polovice $BCEF$.

Pravac CE je okomica u točki D na pravac OD a točke C i E su presjeci pravca CE s pravcima AC i EG .

```
pCE:=okomica(tD, pravac2t(t0, tD));
tC:=presjek2p(pAC, pCE); tE:=presjek2p(pEG, pCE);
```

Površina desne polovice $BCEF$ je zbroj površina trokuta BCE i BEF .

```
povrsina3t(tB, tC, tE)+povrsina3t(tB, tE, tF);
```


Program Maple V će nam za tu sumu izračunati vrijednost $\frac{2R^2}{\cos \theta}$. Kako je $b = 2R \cos \theta$ tražena je površina trapeza $ACEG$ doista $\frac{8R^3}{b}$. \square

Napomena 2. U zbirci [2] krivo se tvrdi da je površina trapeza $\frac{4R^3}{b}$.

Koristeći pristup iz rješenja zadatka 6 u nastavku ovog članka u sljedećem broju MFL-a (tj. zadatka 1112 u [2]) može se u potpunosti izbjeći upotreba trigonometrijskih funkcija. To rješenje prepuštamo čitateljima.

REFERENCE

- [1] M. Bator, Z. Čerin i M. Čulav, *Analitička geometrija ravnine računalom*, Matematičko-fizički list, (prijavljeno).
- [2] Boris Pavković i Darko Veljan, *Matematika 1 – Zbirka zadataka s uputama i rješenjima za prvi razred srednjih škola (14. dopunjenje izdanje)*, Školska knjiga, Zagreb, 1999.

KOPERNIKOVA 7, 10020 ZAGREB, HRVATSKA, EUROPA
E-mail address: cerin@math.hr